On Possible Use of Quantifier Elimination Software in Upper Secondary Mathematics Education

Yosuke Sato
ysato@rs.kagu.tus.ac.jp

Ryoya Fukasaku
fukasaku@rs.tus.ac.jp

Department of Applied Mathematics
Tokyo University of Science
1-3,Kagurazaka,Shinjukuku, Tokyo, Japan

Abstract

Quantifier elimination (QE) is a powerful tool of computer algebra systems. It enables us to solve many mathematical problems in the areas of science, engineering, economics and education, etc. In this paper, we introduce our attempt to apply QE software of computer algebra systems in upper secondary mathematics education. We focus especially on the education of logical reasoning.

1 Introduction

Quantifier elimination (QE) is a powerful tool for solving polynomial problems over real numbers. By the revolutionary CAD (Cylindrical Algebraic Decomposition) algorithm for QE introduced in [3] together with many successive works for its improvement, we now have several fast QE implementations of computer algebra systems such as Mathematica [9], Maple [5, 6], Reduce [11] and QEPCAD [10]. We can use them for solving many mathematical problems in the areas of science, engineering, economics and education, etc.

In the area of education, several applications of QE software are reported. For example, Todai Robot Project [2] of National Institute of Informatics(Japan) develops an automatic solver for mathematics problems of university entrance examinations [1]. In [7], we introduce an application of QE software for analyzing elementary geometry problems of upper secondary school level such as tough problems given in International Mathematical Olympiad. These applications, however, focus mainly on "solving" a problem. They do not take care of "teaching" the underlying essential contents of the problem.

The logical structure of a problem often has a great effect on its solutions. If we understand the underlying logical structure of a problem, we may be able to have a simple solution. In this paper, we show that QE software is also useful for enhancing student's understanding for the underlying logical structure of a problem.

The paper is organized as follows. In section 2, we give a quick review of QE. In section 3, we show a typical usage of QE software for solving a problem using an entrance examination problem of Tokyo University. The main contribution of the paper is given in section 4. We use an entrance examination problem of Tokyo University of Science as a typical example problem whose underlying logical structure strongly affects its solutions. We show that QE software is also useful for teaching such a logical structure of a problem. Throughout of the whole paper we mainly use Mathematica's QE implementations **Resolve** and **Reduce**. Some computation examples by other QE implementations are also given in section 5.

2 Quantifier Elimination

Quantifier elimination(QE) means the following procedure:

> *For a given first order formula, compute an equivalent quantifier free formula by eliminating all quantifiers.*

We can treat QE in many types of domains. In this paper, however, we deals with QE only in the domain of real numbers, where any first order formula consists from atomic formulas of polynomial equations and inequalities with real coefficients. There are several fast QE implementations in the computer algebra systems Mathematica, Maple, Reduce and QEPCAD. The following examples are QE computations using Mathematica's QE implementations **Resolve** and **Reduce**. **Resolve** simply eliminates quantifiers, whereas **Reduce** computes the equivalent quantifier free formula in a form of cylindrical algebraic decomposition(CAD). Roughly speaking, CAD is a decomposition of a subspace $S = \{(x_1, \ldots, x_n) \in \mathbb{R}^n : (x_1, \ldots, x_n) \in S\}$ of $\mathbb{R}^n$ using inequalities in a form of $f(x_1, \ldots, x_{i-1}) < x_i < g(x_1, \ldots, x_{i-1})$ for $i = 1, \ldots, n$. CAD strongly depends on the order of variables $x_1, \ldots, x_n$. In the following examples, $x_1 = \mathtt{b}, x_2 = \mathtt{a}$ in `In[2]` and $x_1 = \mathtt{a}, x_2 = \mathtt{b}$ in `In[3]`.

```
In[1]:= Resolve[Exists[x, x^2 + a x + b == 0], Reals]

Out[1]:= -a^2 + 4 b <= 0

In[2]:= Reduce[Exists[x, x^2 + a x + b == 0], {a, b}, Reals]

Out[2]:= b <= a^2/4

In[3]:= Reduce[Exists[x, x^2 + a x + b == 0], {b, a}, Reals]

Out[3]:= b <= 0 || (b > 0 && (a <= -2 Sqrt[b] || a >= 2 Sqrt[b]))

In[4]:= F[x_]:= x^3 - x + 1

In[5]:= Reduce[Exists[epsilon,
   epsilon > 0 &&
    ForAll[x1, -epsilon < x - x1 < 0 || 0 < x - x1 < epsilon,
     F[x] > F[x1]]] && M == F[x], {x, M}, Reals]

Out[5]:= x == -(1/Sqrt[3]) && M == 1 + 2/(3 Sqrt[3])
```

3 Solving Problems by QE Software

In this section we show how we can use QE software for solving elementary geometry problems of upper secondary school level using the following problem given in the entrance examination of Tokyo University in 2016 as a typical such example.

Problem No.6 Entrance Examination of Tokyo University 2016

A line segment AB with length 2 moves in the three-dimensional (x, y, z)-coordinate real space satisfying the following two conditions.

(a) The point A lies on the plain $z = 0$.
(b) The point $\mathrm{C}(0, 0, 1)$ lies on the line segment AB.

Let K be the region of the coordinate space consisting of all points P lying in the line segment AB. Let K' be the intersection of K and the upper space $1 \leq z$, i.e. $K' = \{(x, y, z) \in \mathbb{R}^3 : 1 \leq z \text{ and } (x, y, z) \in K\}$. Compute the volume of K'.

Let $(a_1, a_2, 0)$ be the coordinate of A and (b_1, b_2, b_3) be the coordinate of B. We have the following polynomial representations.

(1) The length of AB $= 2 \Leftrightarrow (a_1 - b_1)^2 + (a_2 - b_2)^2 + b_3^2 = 4$.

(2) The point $\mathrm{C}(0, 0, 1)$ lies on the line segment AB
$\Leftrightarrow \exists t(0 \leq t \leq 1)\ \overrightarrow{OC} = \overrightarrow{OA} + t\overrightarrow{AB}$
$\Leftrightarrow \exists t(0 \leq t \leq 1)\ (0, 0, 1) = (a_1, a_2, 0) + t(b_1 - a_1, b_2 - a_2, b_3)$
$\Leftrightarrow \exists t(0 \leq t \leq 1)\ 0 = a_1 + t(b_1 - a_1)$ and $0 = a_2 + t(b_2 - a_2)$ and $1 = tb_3$.

(3) The point $\mathrm{P}(x, y, z)$ lies on the line segment AB
$\Leftrightarrow \exists u(0 \leq u \leq 1)\ \overrightarrow{OP} = \overrightarrow{OA} + u\overrightarrow{AB}$
$\Leftrightarrow \exists u(0 \leq u \leq 1)\ (x, y, z) = (a_1, a_2, 0) + u(b_1 - a_1, b_2 - a_2, b_3)$
$\Leftrightarrow \exists u(0 \leq u \leq 1)\ x = a_1 + u(b_1 - a_1)$ and $y = a_2 + u(b_2 - a_2)$ and $z = ub_3$.

So, we have the following representation of K' in terms of a first order formula:

$$\begin{aligned}
&(x, y, z) \in K' \Leftrightarrow \\
&\quad \exists a_1, a_2, b_1, b_2, b_3, t, u\ (0 \leq t \leq 1 \wedge 0 \leq u \leq 1 \wedge 1 \leq z \wedge \\
&\qquad (a_1 - b_1)^2 + (a_2 - b_2)^2 + b_3^2 = 4 \wedge 0 = a_1 + t(b_1 - a_1) \wedge 0 = a_2 + t(b_2 - a_2) \wedge 1 = tb_3 \wedge \\
&\qquad x = a_1 + u(b_1 - a_1) \wedge y = a_2 + u(b_2 - a_2) \wedge z = ub_3).
\end{aligned}$$

The following picture is the execution of the QE program of Mathematica for this first order formula to obtain its equivalent quantifier free formula in a form of CAD which is a suitable form for definite integration.

```
In[1]:= Reduce[Resolve[Exists[{a1,a2,b1,b2,b3,t,u},
        0<=t<=1&&0<=u<=1&&1<=z&&
        (a1-b1)^2+(a2-b2)^2+b3^2==4&&
        0==a1+t*(b1-a1)&&0==a2+t*(b2-a2)&&1==t*b3&&
        x==a1+u*(b1-a1)&&y==a2+u*(b2-a2)&&z==u*b3],Reals],{z,y,x},Reals]

Out[1]:=
 (z==1&&y==0&&x==0)||
 (1<z<2&&
```

```
  ((y==-Sqrt[((4-8z+3z^2+2z^3-z^4)/z^2)]&&x==0)||
   (-Sqrt[((4-8z+3z^2+2z^3-z^4)/z^2)]<y<Sqrt[(4-8z+3z^2+2z^3-z^4)/z^2]&&
      -Sqrt[((4-8z+3z^2-y^2z^2+2z^3-z^4)/z^2)]<=x<=Sqrt[(4-8z+3z^2-y^2z^2+2z^3-z^4)/z^2])||
   (y==Sqrt[(4-8z+3z^2+2z^3-z^4)/z^2]&&x==0)))||
 (z==2&&y==0&&x==0)
```

From this formula, we can compute the volume of K' using a program of definite integration as follows.

```
In[2]:= Assuming[1<z<2,
          Integrate[Integrate[2Sqrt[(4-8z+3z^2-y^2z^2+2z^3-z^4)/z^2],
            {y,-Sqrt[((4-8z+3z^2+2z^3-z^4)/z^2)],Sqrt[(4-8z+3z^2+2z^3-z^4)/z^2]}],
            {z,1,2}]]

Out[2]:= Pi (17/3 - 8 Log[2])
```

4 Teaching Logic by QE Software

As is shown in the previous section, QE software has great power for automatic solving of elementary geometry problems. Consider the following problem given in the entrance examination of Tokyo University of Science in 2016 which is much easier than the above problem.

Problem No.1 Entrance Examination of Tokyo University of Science 2016

(1) Compute the range of real values of a such that the quadratic equation $x^2 + (a+1)x + a^2 + a - 1 = 0$ with one unknown x has a real root.

(2) Compute the range of real numbers which consists of all possible real roots of the above quadratic equation for real values of a in the range computed in (1).

Using the QE program **Resolve** of Mathematica, we can also give solutions for these problems as follows.

```
In[3]:= Resolve[Exists[x,x^2+(a+1)x+a^2+a-1==0],Reals]

Out[3]:= -(5/3)<=a<=1

In[4]:= Resolve[Exists[a,-(5/3)<=a<=1&&x^2+(a+1)x+a^2+a-1==0],Reals]

Out[4]:= -(5/3)<=x<=1
```

The most important point for the solution of (2) is that we have the following equivalent relation:

$$\exists a(-\frac{5}{3} \le a \le 1 \land x^2 + (a+1)x + a^2 + a - 1 = 0) \Leftrightarrow \exists a(x^2 + (a+1)x + a^2 + a - 1 = 0).$$

Once we notice it, we can immediately get the solution $\frac{5}{3} \le x \le 1$ of (2) by the symmetric property of the polynomial $x^2 + (a+1)x + a^2 + a - 1$ without doing any computation.
The above executions of the QE program do not give any information concerning this underlying logical structure of the problem.

In the following, we provide two typical solutions for (2). The first one is by a student, who does not notice the above logical structure and writes an awfully long (but correct) solution. The second one is a model answer given by one of the most major cram schools in Japan. It gives a too smart solution, there is no description why we have the above equivalent relation but it simply says "**Clearly it is equivalent to ...**".

A solution of (2) by some student

Let $f(x,a) = a^2 + (x+1)a + x^2 + x - 1$. The range consists of all real values of x such that the graph $y = f(x,a)$ in the (a,y)-coordinate plane intersects the a-axis at $-\frac{5}{3} \leq a \leq 1$. There are five possible cases:

Case 1. The graph intersects once at $-\frac{5}{3} < a < 1$
and the intersection point is not a point of tangency.

It is equivalent to $f(x, -\frac{5}{3})f(x,1) < 0 \Leftrightarrow (3x-1)^2(x+1)^2 < 0$.
There no real values for x satisfying it.

Case 2. The graph intersects twice at $-\frac{5}{3} < a < 1$.
Case 3. The graph intersects once at $-\frac{5}{3} < a < 1$
and the intersection point is a point of tangency.

For either case 2 or 3, an equivalent condition is
$-\frac{5}{3} < -\frac{x+1}{2} < 1$(the axis of the parabolic graph $y = f(x,a)$ is between $-\frac{5}{3}$ and 1.)
$\wedge\ f(x, -\frac{5}{3}) > 0 \wedge f(x,1) > 0$
$\wedge\ (x+1)^2 - 4(x^2+x-1) \geq 0$(the discriminant ≥ 0).
It is equivalent to $-\frac{5}{3} \leq x < -1 \vee -1 < x < \frac{1}{3} \vee \frac{1}{3} < x \leq 1$.

Case 4. The graph intersects at $a = -\frac{5}{3}$.

It is equivalent to $f(-\frac{5}{3}, x) = 0 \Leftrightarrow x = \frac{1}{3}$.

Case 5. The graph intersects at $a = 1$.

It is equivalent to $f(1,x) = 0 \Leftrightarrow x = -1$.

Consequently the desired range is $-\frac{5}{3} \leq x \leq 1$.

A model solution of (2) by some major cram school

Let $f(x,a) = a^2 + (x+1)a + x^2 + x - 1$. The range consists of all real values of x such that $\exists a(-\frac{5}{3} \leq a \leq 1 \wedge f(x,a) = 0)$. **Clearly it is equivalent to** $\exists a f(x,a) = 0$. Since the polynomial $f(x,a)$ is symmetric about x and a, the desired range is $-\frac{5}{3} \leq x \leq 1$.

For the equivalent relation:

$$\exists a(-\frac{5}{3} \leq a \leq 1 \wedge f(x,a) = 0) \Leftrightarrow \exists a f(x,a) = 0$$

how can we provide a clear deduction ?

The most essential way is the inference of first-order predicate logic.
Define the formulas $F(x,a)$ and $G(a)$ by $F(x,a) \Leftrightarrow f(x,a) = 0$ and $G(a) \Leftrightarrow -\frac{5}{3} \leq a \leq 1$

By the solution of (1), we know that the following sentence holds:

$$\forall a(\exists x F(x,a) \Leftrightarrow G(a))$$

Furthermore, we have the following sequence of valid sentences. $\Downarrow$ denotes that the upper sentence implies the lower sentence, $\Updownarrow$ denotes that the upper sentence and the lower sentence are equivalent.

$$\forall a(\exists x F(x,a) \Leftrightarrow G(a))$$
$$\Downarrow$$
$$\forall a(\forall x(F(x,a) \Rightarrow G(a)))$$
$$\Updownarrow$$
$$\forall x(\forall a(F(x,a) \Rightarrow G(a)))$$
$$\Updownarrow$$
$$\forall x(\forall a(F(x,a) \Leftrightarrow F(x,a) \wedge G(a)))$$
$$\Downarrow$$
$$\forall x(\exists a F(x,a) \Leftrightarrow \exists a(F(x,a) \wedge G(a)))$$

As a consequence, the last sentence holds, i.e. we have the desired equivalent relation:

$$\forall x(\exists a f(x,a) = 0 \Leftrightarrow \exists a(f(x,a) = 0 \wedge -\frac{5}{3} \leq a \leq 1)).$$

QE software is also useful for enhancing student ' s understanding for such logical inferences. Mathematica's QE software **Resolve** returns `True` to the above sentences as in the following picture.

```
In[1]:= F[x_,a_] := x^2+(a+1)*x+a^2+a-1==0
In[2]:= G[a_,b_,c_] := b<=a<=c

In[3]:= Resolve[ForAll[a,Equivalent[Exists[x,F[x,a]],G[a,-5/3,1]]],Reals]
Out[3]:= True

In[4]:= Resolve[ForAll[{x,a},Implies[F[x,a],G[a,-5/3,1]]],Reals]
Out[4]:= True

In[5]:= Resolve[ForAll[{x,a},Equivalent[
           F[x,a],F[x,a]&&G[a,-5/3,1]]],Reals]
Out[5]:= True

In[6]:= Resolve[ForAll[x,Equivalent[
          Exists[a,F[x,a]],Exists[a,F[x,a]&&G[a,-5/3,1]]]],Reals]
Out[6]:= True
```

Unfortunately, these executions do not help students to understand the logical structure of the problem. Actually we do not need QE software for understanding correctness of the above sequence of sentences since they are simple and easy to understand.

However, QE software have a great effect on teaching students that the directions of the above symbols $\Downarrow$ are only one side. The first $\Downarrow$ is only one side since we can obviously replace $-\frac{5}{3} \leq a \leq 1$ by $b \leq a \leq c$ with any real numbers b, c such that $b \leq -\frac{5}{3}$ and $c \geq 1$.

Using QE software, we can check that it is also a necessary condition as follows.

```
In[7]:= Reduce[ForAll[a,Implies[Exists[x,F[x,a]],G[a,b,c]]],{c,b},Reals]

Out[7]:= c >= 1 && b <= -(5/3)
```

The second $\Downarrow$ is only one side too because we can also replace $-\frac{5}{3} \leq a \leq 1$ by $b \leq a \leq c$ with any real numbers b, c such that $b \leq -\frac{5}{3}$ and $c \geq 1$. Remarkably, in this case, it is not a necessary condition. We can also compute the necessary and sufficient condition by QE software.

```
In[8]:= Reduce[ForAll[x,Equivalent[
         Exists[a,F[x,a]],Exists[a,F[x,a]&&G[a,b,c]]]],{c,b},Reals]

Out[8]:= (1/3<=c<=1&&b<=1/2(-1-c)-1/2Sqrt[5-2c-3c^2])||(c>1&&b<=-1)
```

The meaning of this formula can be further explored by using a graphical interface of a computer algebra system such as the following picture of Mathematica.

```
In[9]:= P[c_,b_] :=(1/3<=c<=1&&b<=1/2(-1-c)-1/2Sqrt[5-2c-3c^2])||(c>1&&b<=-1)
In[10]:= Show[ContourPlot[{-1+a+a^2+(1+a)x+x^2==0},{x,-2,1.2},{a,-2,1.2},
  PlotRange->{{-2,1.2},{-2,1.2}},AspectRatio->Automatic],
  RegionPlot[P[x,a],{x,-2,1.2},{a,-2,1.2}]]
```



The dark blue region is the region $\{(c, b) \in \mathbb{R}^2 : \forall x(\exists a f(x, a) = 0 \Leftrightarrow \exists a f(x, a) = 0 \wedge b \leq a \leq c)\}$.

One of the authors has used the above example in delivery lectures at several high schools for teaching underlying advanced contents of mathematics problems of the entrance examinations of our university. According to questionnaire research, many students feel that QE software helps them to understand logical reasoning.

5 Computations by Other QE Implementations

We can use other QE implementations. There are four major QE implementations of computer algebra systems except for Mathematica's QE programs.The most classical one is QEPCAD [10]. Though the program is not so fast as the other implementations in general, it produces very compact output formulas. QEPCAD is a freeware software. We have an another free QE program **rlqe** implemented as a part of redlog developed in the computer algebra system Reduce. Its output formulas are much more complicated than the others in general. There are two QE implementations running on the computer algebra system Maple. One is SynRAC [5] and the other is Real Quantifier Elimination in the RegularChains Library [6]. SynRAC has a very fast QE program in general and returns reasonably compact output formulas by using several simplification techniques. QE computations of the RegularChain Library is based on the computation of regular chains. Its output formulas tend to be more complicated than QEPCAD and SynRAC in general.

In the following we show the log files of QE computations by those implementations for the formula of `In[8]` treated in the previous section. Since some program can handle only prenex normal forms, we use its equivalent prenex normal form as an input formula. The output formula of **rlqe** is extremely long and complicated, so we omit it.

QEPCAD

```
=======================================================
                Quantifier Elimination
                          in
            Elementary Algebra and Geometry
                          by
      Partial Cylindrical Algebraic Decomposition

               Version B 1.69, 16 Mar 2012

                          by
                      Hoon Hong
                  (hhong@math.ncsu.edu)

With contributions by: Christopher W. Brown, George E.
Collins, Mark J. Encarnacion, Jeremy R. Johnson
Werner Krandick, Richard Liska, Scott McCallum,
Nicolas Robidoux, and Stanly Steinberg
=======================================================
Enter an informal description  between '[' and ']':
[]
Enter a variable list:
(b,c,x,d,a)
Enter the number of free variables:
2
Enter a prenex formula:
(A x)(E d)(A a)[x^2+(a+1) x+a^2+a-1 /= 0 \/ [x^2+(d+1) x+d^2+d-1=0 /\ b>=d /\ d >=c]].
=======================================================

Before Normalization >
finish

An equivalent quantifier-free formula:

3 b - 1 >= 0 /\ c + 1 <= 0 /\ c^2 + b c + c + b^2 + b - 1 >= 0
```

rlqe of Reduce

```
Reduce (Free CSL version), 04-Aug-11 ...

1: load_package redlog;
2: rlset R;
3: phi := all(x,ex(d,all(a,(a^2+a*x+a+x^2+x-1<>0 or (d^2+d*x+d+x^2+x-1=0
          and b<=d and d<=c)))));
4: rlqe phi;
```

SynRAC

```
    |\^/|     Maple 2016 (X86 64 LINUX)
._|\|   |/|_. Copyright (c) Maplesoft, a division of Waterloo Maple Inc. 2016
 \  MAPLE  /  All rights reserved. Maple is a trademark of
 <____ ____>  Waterloo Maple Inc.
      |       Type ? for help.
> libname := libname, "/home/ysato/synrac/":
> with(SyNRAC):

> phi:=All([x],Ex([d],All([a],Or(x^2+(a+1)*x+a^2+a-1<>0,
  And(x^2+(d+1)*x+d^2+d-1=0,b<=d,d<=c))))):

> qe(phi);
                                         2          2
 And(b <= -1, b - c < 0, -3 c <= -1, -b  - b c - c  - b - c <= -1)
```

Regular Chain Library

```
    |\^/|     Maple 2016 (X86 64 LINUX)
._|\|   |/|_. Copyright (c) Maplesoft, a division of Waterloo Maple Inc. 2016
 \  MAPLE  /  All rights reserved. Maple is a trademark of
 <____ ____>  Waterloo Maple Inc.
      |       Type ? for help.
> libname := ‘‘/home/ysato/RegularChain’’, libname:
> with(RegularChains):
> with(ConstructibleSetTools):
> with(SemiAlgebraicSetTools):
> phi:=‘&A‘([x]),‘&E‘([d]),‘&A‘([a]),‘&or‘(x^2+(a+1)*x+a^2+a-1<>0,‘&and‘(
x^2+(d+1)*x+d^2+a1-1=0,b<=d,d<=c)):
> QuantifierElimination(phi);

‘&or‘(‘&or‘(‘&or‘(‘&or‘(‘&or‘(‘&or‘(‘&or‘(‘&or‘(‘&or‘(‘&or‘(‘&or‘(‘&or‘(‘&or‘(
‘&and‘(3*c-1 = 0,3*b < -5),‘&and‘(3*c-1 = 0,3*b+5 = 0)),‘&and‘(c-1 = 0,b+1 = 0)
),‘&and‘(c-1 = 0,3*b < -5)),‘&and‘(c-1 = 0,3*b+5 = 0)),‘&and‘(‘&and‘(c-1 = 0,b
< -1),0 < 3*b+5)),‘&and‘(0 < c-1,b+1 = 0)),‘&and‘(0 < c-1,3*b < -5)),‘&and‘(0 <
c-1,3*b+5 = 0)),‘&and‘(‘&and‘(0 < c-1,b < -1),0 < 3*b+5)),‘&and‘(‘&and‘(0 < 3*c
-1,c < 1),3*b < -5)),‘&and‘(‘&and‘(0 < 3*c-1,c < 1),3*b+5 = 0)),‘&and‘(‘&and‘(
‘&and‘(0 < 3*c-1,c < 1),b < -1),b^2+b*c+c^2+b+c-1 = 0)),‘&and‘(‘&and‘(‘&and‘(
‘&and‘(0 < 3*c-1,c < 1),b < -1),0 < b^2+b*c+c^2+b+c-1),0 < 3*b+5))
```

6 Conclusion and Remarks

Remember that "valid" means that the sentence holds for any interpretation. Therefore the sequence of the sentences given in section 4 hold for any interpretation of $F(x, a)$ and $G(a)$. Unfortunately, QE computation cannot handle validity of a sentence. We need a theorem prover of first-order predicate logic such as [4, 8]. For education of upper secondary mathematics, however, we do not think that students encounter a complicated valid sentence such that we need to use a theorem prover for checking its validity. What we actually need is not checking the validity of a sentence but understanding the logical structure in a concrete domain, it is the domain of real numbers we discuss in this paper. As is shown in section 4, QE computation is useful for such a purpose. Any theorem prover of first-order predicate logic cannot handle QE computations presented in section 4.

References

[1] Arai, N.H., Matsuzaki, T., Iwane, H. and Anai, H.: Mathematics by Machine, Proceedings of International Symposium on Symbolic and Algebraic Computation, pp. 1-8, ACM, 2014.

[2] Todai Robot Project. `http://21robot.org/research_activities/math/`

[3] Collins, G, E.: Quantifier elimination for real closed fields by cylindrical algebraic decomposition. Automata theory and formal languages (Second GI Conf., Kaiserslautern, 1975), Lecture Notes in Comput. Sci., Vol. 33, pp. 134-183, Springer, Berlin, 1975.

[4] E: a theorem prover for full first-order logic with equality.
`http://wwwlehre.dhbw-stuttgart.de/~sschulz/E/E.html`

[5] SyNRAC a software package for quantifier elimination (QE).
`http://www.fujitsu.com/jp/group/labs/en/resources/tech/announced-tools/synrac/`

[6] The RegularChains Library.
`http://www.regularchains.org/index.html`

[7] Sato, Y. and Fukasaku, R. Detecting unnecessary assumptions of elementary geometry problems by CAS. Proceedings of the 20th Asian Technology Conference in Mathematics, pp. 316-325, 2015.

[8] Prover9: an automated theorem prover for first-order and equational logic.
`http://www.cs.unm.edu/~mccune/prover9/`

[9] Mathematica Tutorial: tutorial/ComplexPolynomialSystems
Mathematica Tutorial: tutorial/RealPolynomialSystems

[10] QEPCAD-Quantifier Elimination by Partial Cylindrical Algebraic Decomposition.
`http://www.usna.edu/CS/qepcadweb/B/QEPCAD.html`

[11] Redlog: An integral part of the interactive computer algebra system Reduce.
`http://www.redlog.eu/`